ExteriorDerivative

`ExteriorDerivative[`*expr*`]`
represents the exterior derivative of *expr*.

Details

- The exterior derivative template is on the ∇ tab of the Common Operations palette. **d**[□].
- Exterior derivatives are evaluated with `EvaluateExteriorDerivatives`.
- Exterior derivatives operate on the coordinates.

Examples (2)

Basic Examples (1)

In[1]:=
```
<< GrassmannCalculus`
```

In[2]:=
```
SetEuclideanNSpace[3, {x, y, z}, "Form"]
```

The exterior derivative the the coordinates are the form basis vectors.

In[3]:=
```
ExteriorDerivative /@ GrassmannCoordinates
% // EvaluateExteriorDerivatives
```

Out[3]= {**d**[x], **d**[y], **d**[z]}

Out[3]= {dx, dy, dz}

An exterior product operating on a function.

In[4]:= **d**[$3\,x\,y^2$ Sin[z]]
```
% // EvaluateExteriorDerivatives
```

Out[4]= **d**[$3\,x\,y^2$ Sin[z]]

Out[4]= $3\,dz\,x\,y^2$ Cos[z] + 6 dy x y Sin[z] + $3\,dx\,y^2$ Sin[z]

Exterior derivatives can be linearly expanded before evaluation

In[5]:= **d**[$a\,x + 3\,y\,z^2 + z\,dx$]
```
% // ExpandExteriorDerivatives
% // EvaluateExteriorDerivatives
```

Out[5]= **d**[$a\,x + dx\,z + 3\,y\,z^2$]

Out[5]= a **d**[x] + **d**[dx z] + 3 **d**[$y\,z^2$]

Out[5]= $a\,dx + 6\,dz\,y\,z + 3\,dy\,z^2 - dx \wedge dz$

©1988–2015 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

Axioms for the exterior derivative (1)

Define two coordinate systems.

In[1]:=
```
SetEuclideanNSpace[3, {x, y, z}, "Form"];
⋆⋆S[(f | Fu | Fv | Fw)[__]];
XYZCoordinates = CurrentGrassmannAssociation;
setXYZCoordinates := SetActiveAssociation[XYZCoordinates]
SetEuclideanNSpace[3, {u, v, w}, "Form"];
⋆⋆S[(f | Fu | Fv | Fw)[__]];
UVWCoordinates = CurrentGrassmannAssociation;
setUVWCoordinates := SetActiveAssociation[UVWCoordinates]
setXYZCoordinates
```

These are the axioms of exterior derivatives, taken from John M. Lee, *Introduction to Smooth Manifolds*, 2013, p 364) with examples.

1) The exterior derivative is linear over the real numbers.

In[2]:=
```
d[1.2 x + x dy + a z dx∧dy]
% // ExpandExteriorDerivatives
```

Out[2]= `d[1.2 x + dy x + a z dx∧dy]`

Out[2]= `1.2 d[x] + d[dy x] + a d[z dx∧dy]`

2) If ψ is a k-form, then: $\mathbf{d}[\psi\wedge\omega] == \mathbf{d}[\psi]\wedge\omega + (-1)^k\,\psi\wedge\mathbf{d}[\omega]$.

In[3]:=
```
ψform = dx + x dy - 3 dz;
ωform = y dx;
d[ψform∧ωform] == d[ψform]∧ωform + (-1)^Grade[ψform] ψform∧d[ωform]
% // EvaluateExteriorDerivatives
```

Out[3]= `d[(dx - 3 dz + dy x)∧(dx y)] == -((dx - 3 dz + dy x)∧d[dx y]) + d[dx - 3 dz + dy x]∧(dx y)`

Out[3]= `True`

3) $\mathbf{d}[\mathbf{d}[\square]] == 0$

In[4]:=
```
d[d[x y^2 Sin[z]]]
% // EvaluateExteriorDerivatives
```

Out[4]= `d[d[x y^2 Sin[z]]]`

Out[4]= `0`

In[5]:=
```
d[d[y dx∧dz]]
% // EvaluateExteriorDerivatives
```

Out[5]= `d[d[y dx∧dz]]`

Out[5]= `0`

4) The exterior derivative commutes with pullbacks.

Define a generic mapping from {x, y, z} coordinates to {u, v, w} coordinates and the coordinate pullback.

In[6]:=
```
F = {x, y, z} ↦ {Fu[x, y, z], Fv[x, y, z], Fw[x, y, z]}
uvwPullback = {u, v, w} → F[x, y, z] // Thread
```

Out[6]= `Function[{x, y, z}, {Fu[x, y, z], Fv[x, y, z], Fw[x, y, z]}]`

Out[6]= `{u → Fu[x, y, z], v → Fv[x, y, z], w → Fw[x, y, z]}`

©1988–2015 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

We need to use the `PullbackForms` routine but lets define a shortcut form tailored to our specific example so as to more compactly display the axiom.

In[7]:=
```
Pull[form_] := PullbackForms[setUVWCoordinates, setXYZCoordinates, uvwPullback][form];
```

We can then simply test the axiom.

In[8]:=
```
ωform = u v^2 e^w;
(setUVWCoordinates; Pull[d[ωform] // EvaluateExteriorDerivatives]) ==
  (setXYZCoordinates;
   d[Pull[ωform]] // EvaluateExteriorDerivatives) // Simplify
```

Out[8]=
```
True
```

See Also

ExpandExteriorDerivatives · **EvaluateExteriorDerivatives** · **VectorOperator** · **DirectionalDerivative** · **LieDerivative**

Related Guides

- Calculus
- Contractors
- Derivatives
- Grassmann Calculus

©1988–2015 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language